/* Begin_Procs taCreateEmployeeTax2020 */
if exists (select * from dbo.sysobjects where id = Object_id('dbo.taCreateEmployeeTax') and type = 'P')
begin
	drop proc dbo.taCreateEmployeeTax
end
go

create procedure dbo.taCreateEmployeeTax
/*
**********************************************************************************************************
* (c) 2003 Microsoft Business Solutions, Inc.
**********************************************************************************************************
*
* PROCEDURE NAME:	taCreateEmployeeTax
*
* SANSCRIPT NAME:	NA
*
* PARAMETERS:
*
*
* DESCRIPTION:
*		TA Create Employee Taxes Stored Procedure
*
*
*
* TABLES:
*		Table Name			Access
*		==========			======
*		UPR00300			Read/Write
*		UPR00100			Read
*		UPR41100			Read
*		UPR41400			Read
*
*
* PROCEDURES CALLED:
*			taUpdateString
*			tasmGetNextNoteIndex
*			taCreateEmployeeTaxPre
*			taCreateEmployeeTaxPost
*
*
* DATABASE:
*		Company
*
*
* RETURN VALUE:
*
*	0 = Successful
*	non-0 = Not successful
*
* REVISION HISTORY:
*
*	Date		Who		Comments
*	------------- 	--------        -------------------------------------------------
*	04/02/02	ggeiszle	Initial Creation
*	01/28/02	jgould		Modified Grants original code to get it to R7.0
*	02/27/04	Jeff Gould	Updated to Release 8.0 - No Changes Required
*	06/16/05	Jeff Gould	Updated to Release 9.0 - No Changes Required
*	09/11/07	Jeff Gould	Fixed Bug ID#41786
*	04/29/12	Tim Satrom	Fixed Bug ID#65159
*
**********************************************************************************************************
*
**********************************************************************************************************
*/	
@I_vEMPLOYID char(15),			/* Employee ID - <Required> 										*/
@I_vFDFLGSTS char(6)='',		/* Federal Filing Status - <Optional> - MAR=Married, SINGLE=Single, BLANK=exempt, NRA=Nonresident Alien		*/
@I_vFEDEXMPT smallint=0,		/* Number of exemptions claimed - <Optional>								*/
@I_vADFDWHDG numeric(19,5)=0,		/* Additional Federal Withholding - <Optional> 								*/
@I_vESTFEDWH numeric(19,5)=0,		/* Estimated Federal Withholding - <Optional>								*/
@I_vSTATECD char(2)='',			/* State Code - <Optional>  										*/
@I_vLOCALTAX char(6)='',		/* Local Tax Code - <Optional> 										*/
@I_vW2BF942E tinyint=0,			/* W2 Box for 942 Employee - <Optional> - 0=False&1=True						*/
@I_vW2BFDCSD tinyint=0,			/* W2 Box for Deceased - <Optional> - 0=False&1=True							*/
@I_vW2BFDCMP tinyint=0,			/* W2 Box for Deferred Compensation - <Optional> - 0=False&1=True					*/
@I_vW2BFLREP tinyint=0,			/* W2 Box for Legal Representative - <Optional> - 0=False&1=True					*/
@I_vW2BFPPLN tinyint=0,			/* W2 Box for Pension Plan - <Optional> - 0=False&1=True						*/
@I_vW2BFSTEM tinyint=0,			/* W2 Box for Statutory Employee - <Optional> - 0=False&1=True						*/
@I_vMCRQGEMP tinyint=0,			/* Medicare Qualified Government Employee - <Optional>							*/
@I_vEICFLGST char(6)='',		/* EIC Filing Status - <Optional> - BLANK = Not Eligible						*/
					/* MAR=Married Both Spouses file Cert, 									*/
					/* MAR2=Married without Spouse filing Cert, & 								*/
					/* SINGLE=Single/Head of Household									*/
@I_vNYTXDiff tinyint=0,			/* Withhold NY Tax Difference - <Optional>								*/
@I_vUpdateIfExists tinyint=1,		/* Flag to allow Employe Tax data to be updated if it exists.  0=no, 1=yes					*/
@I_vRequesterTrx smallint=0,		/* Requester Transaction <Optional> - 0=false,1=true - if true than populates Requester shadow table	*/
@I_vUSRDEFND1 char(50) = '',	    	/* User Defined field - developer use only								*/
@I_vUSRDEFND2 char(50) = '',	    	/* User Defined field - developer use only								*/
@I_vUSRDEFND3 char(50) = '',	    	/* User Defined field - developer use only								*/
@I_vUSRDEFND4 varchar(8000) = '',	/* User Defined field - developer use only								*/
@I_vUSRDEFND5 varchar(8000) = '',	/* User Defined field - developer use only								*/
@O_iErrorState int output,		/*Return value:  0=No Errors, 1=Error Occurred								*/
@oErrString varchar(255) output		/*Return Error Code List										*/

with encryption
as

set transaction isolation level read uncommitted
set nocount on

declare
	@sCompanyID int,
	@iCustomState int,
	@iStatus int,
	@O_oErrorState int,
	@nNextNoteIndex	numeric(19,5),		/* Stores the next note index						*/
	@iGetNextNoteIdxErrState int,		/* Return code from call to proc to get next note index			*/
	@iError int,				/* Error code from system calls						*/
	@DBName char(50),			/* Database name of the company						*/
	@iAddCodeErrState int

select
	@O_iErrorState = 0,			/* Set error state to 0 (no errors found)*/
	@O_oErrorState = 0,
	@sCompanyID = CMPANYID from DYNAMICS..SY01500 where INTERID = db_name()

if (@oErrString is NULL)
begin
	select @oErrString = ''
end

/*********************** Call Pre Proc *****************************************************/
exec @iStatus = taCreateEmployeeTaxPre
	@I_vEMPLOYID output,
	@I_vFDFLGSTS output,
	@I_vFEDEXMPT output,
	@I_vADFDWHDG output,
	@I_vESTFEDWH output,
	@I_vSTATECD output,
	@I_vLOCALTAX output,
	@I_vW2BF942E output,
	@I_vW2BFDCSD output,
	@I_vW2BFDCMP output,
	@I_vW2BFLREP output,
	@I_vW2BFPPLN output,
	@I_vW2BFSTEM output,
	@I_vMCRQGEMP output,
	@I_vEICFLGST output,
	@I_vNYTXDiff output,
	@I_vUpdateIfExists output,
	@I_vRequesterTrx output,
	@I_vUSRDEFND1 output,
	@I_vUSRDEFND2 output,
	@I_vUSRDEFND3 output,
	@I_vUSRDEFND4 output,
	@I_vUSRDEFND5 output,
	@O_iErrorState = @iCustomState output,
	@oErrString = @oErrString /*@iCustomErrString*/ output
select @iError = @@error
if @iStatus = 0 and @iError <> 0
begin
	select @iStatus = @iError
end
if (@iStatus <> 0) or (@iCustomState <> 0)
begin
	select @O_iErrorState = 4096 /* Pre Custom Business Logic for taCreateEmployeeTaxPre returned an error value */ /*EMPLOYID*/
	exec @iStatus = taUpdateString
 		@O_iErrorState,
 		@oErrString,
 		@oErrString output,
 		@O_oErrorState output
	return (@O_iErrorState)
end

/*********************** Validate input parameters for null *****************************************************/
/* Verify input parameters are not null */
if	(
	 @I_vEMPLOYID is NULL or
	 @I_vFDFLGSTS is NULL or
	 @I_vFEDEXMPT is NULL or
	 @I_vADFDWHDG is NULL or
	 @I_vESTFEDWH is NULL or
	 @I_vSTATECD is NULL or
	 @I_vLOCALTAX is NULL or
	 @I_vW2BF942E is NULL or
	 @I_vW2BFDCSD is NULL or
	 @I_vW2BFDCMP is NULL or
	 @I_vW2BFLREP is NULL or
	 @I_vW2BFPPLN is NULL or
	 @I_vW2BFSTEM is NULL or
	 @I_vMCRQGEMP is NULL or
	 @I_vEICFLGST is NULL or
	 @I_vNYTXDiff is NULL or
	 @I_vUpdateIfExists is NULL or
	 @I_vRequesterTrx is NULL or
	 @I_vUSRDEFND1 is NULL or
	 @I_vUSRDEFND2 is NULL or
	 @I_vUSRDEFND3 is NULL or
	 @I_vUSRDEFND4 is NULL or
	 @I_vUSRDEFND5 is NULL
	)
begin
	/* input variable contains null value */
	select @O_iErrorState = 4594 /* A null was found in at least one input parameter for taCreateEmployeeTax */ /*EMPLOYID*/
	exec @iStatus = taUpdateString
 		@O_iErrorState,
 		@oErrString,
 		@oErrString output,
 		@iAddCodeErrState output
end

/*********************** Master Record Upper Case ***************************************************************/
select	@I_vEMPLOYID =UPPER(@I_vEMPLOYID),
	@I_vFDFLGSTS = UPPER(@I_vFDFLGSTS),
	@I_vEICFLGST = UPPER(@I_vEICFLGST),
	@I_vSTATECD = UPPER(@I_vSTATECD),
	@I_vLOCALTAX = UPPER(@I_vLOCALTAX)

/******************************************Validate Required Fields****************************************/
/* Validate the Employee ID is not blank */
if ( @I_vEMPLOYID = '' )
begin
	select @O_iErrorState = 1569 /* The Employee ID (EMPLOYID) is blank */ /*EMPLOYID*/
	exec @iStatus = taUpdateString
	  	@O_iErrorState,
	  	@oErrString,
	  	@oErrString output,
		@iAddCodeErrState output
end

/* Validate the Employee ID actually exists in the UPR00100 table */
if ( @I_vEMPLOYID <> '' )
begin
	if (not exists (select 1 from UPR00100 (nolock) where EMPLOYID = @I_vEMPLOYID ))
	begin
		select @O_iErrorState = 1570 /* The Employee ID (EMPLOYID) passed does not exist in the Employee Master table - UPR00100 */ /*EMPLOYID*/
		exec @iStatus = taUpdateString
	  		@O_iErrorState,
	  		@oErrString,
	  		@oErrString output,
			@iAddCodeErrState output
	end
end

/* Validate the Number of Exemptions <> < 0 */
if ( @I_vFEDEXMPT < 0 ) 
begin
	select @O_iErrorState = 1571 /* The Number of Exemptions (FEDEXMPT) can not be < 0 */ /*EMPLOYID*/ /*FEDEXMPT*/
	exec @iStatus = taUpdateString
	  	@O_iErrorState,
	  	@oErrString,
	  	@oErrString output,
		@iAddCodeErrState output
end

/* Validate Estimated Federal Withholding is not < 0 */
if ( @I_vESTFEDWH < 0 )
begin
	select @O_iErrorState = 1572 /* The Estimated Federal Withholding (ESTFEDWH) can not be < 0 */ /*EMPLOYID*/ /*ESTFEDWH*/
	exec @iStatus = taUpdateString
	  	@O_iErrorState,
	  	@oErrString,
	  	@oErrString output,
		@iAddCodeErrState output
end

/* Validate the Tax Withholding State is a valid value */
if ( @I_vSTATECD <> '' ) 
begin
	if (not exists (select 1 from UPR41100 (nolock) where STATECD = @I_vSTATECD ))
	begin
		select @O_iErrorState = 1573 /* The State (STATECD) passed does not exist in the Payroll State Code Setup Table - UPR41100 */ /*EMPLOYID*/ /*STATECD*/
		exec @iStatus = taUpdateString
	  		@O_iErrorState,
	  		@oErrString,
	  		@oErrString output,
			@iAddCodeErrState output
	end
end

/* Validate the Local Tax passed exists */
if ( @I_vLOCALTAX <> '' ) 
begin
	if (not exists (select 1 from UPR41400 (nolock) where LOCALTAX = @I_vLOCALTAX ))
	begin
		select @O_iErrorState = 1574 /* The Local Tax (LOCALTAX) passed does not exist in the Payroll Local Tax Setup Table - UPR414100 */ /*EMPLOYID*/ /*LOCALTAX*/
		exec @iStatus = taUpdateString
	  		@O_iErrorState,
	  		@oErrString,
	  		@oErrString output,
			@iAddCodeErrState output
	end
end

/* Validate W2 Box for 942 Employee is a valid value */
if ( @I_vW2BF942E < 0 or @I_vW2BF942E > 1 ) 
begin
	select @O_iErrorState = 1575 /* The W2 Box for 942 Employee (W2BF942E) is invalid - 0 and 1 are valid values */ /*EMPLOYID*/ /*W2BF942E*/
	exec @iStatus = taUpdateString
	  	@O_iErrorState,
	  	@oErrString,
	  	@oErrString output,
		@iAddCodeErrState output
end

/* Validate W2 Box for Deceased Employee is a valid value */
if ( @I_vW2BFDCSD < 0 or @I_vW2BFDCSD > 1 ) 
begin
	select @O_iErrorState = 1576 /* The W2 Box for Deceased Employee (W2BFDCSD) is invalid - 0 and 1 are valid values */ /*EMPLOYID*/ /*W2BFDCSD*/
	exec @iStatus = taUpdateString
	  	@O_iErrorState,
	  	@oErrString,
	  	@oErrString output,
		@iAddCodeErrState output
end

/* Validate W2 Box for Deferred Compensation is valid */
if( @I_vW2BFDCMP < 0 or @I_vW2BFDCMP > 1 )
begin
	select @O_iErrorState = 1577 /* W2 Box for Deferred Compensation (W2BFDCMP) is invalid - 0 and 1 are valid values */ /*EMPLOYID*/ /*W2BFDCMP*/
	exec @iStatus = taUpdateString
		@O_iErrorState,
		@oErrString,
		@oErrString output,
		@iAddCodeErrState output
end

/* Validate W2 Box for Legal Rep is valid */
if ( @I_vW2BFLREP < 0 or @I_vW2BFLREP > 1 )
begin
	select @O_iErrorState = 1578 /* W2 Box for Legal Rep (W2BFLREP) is invalid - 0 and 1 are valid values */ /*EMPLOYID*/ /*W2BFLREP*/
	exec @iStatus = taUpdateString
		@O_iErrorState,
		@oErrString,
		@oErrString output,
		@iAddCodeErrState output
end

/* Validate W2 Box for Pension Plan is valid */
if (@I_vW2BFPPLN < 0 or @I_vW2BFPPLN > 1 )
begin
	select @O_iErrorState = 1579 /* W2 Box for Pension Plan (W2BFPPLN) is invalid - 0 and 1 are valid values */ /*EMPLOYID*/ /*W2BFPPLN*/
	exec @iStatus = taUpdateString
		@O_iErrorState,
		@oErrString,
		@oErrString output,
		@iAddCodeErrState output
end

/* Validate the W2 Box for Statuatory Employee is valid */
if ( @I_vW2BFSTEM < 0 or @I_vW2BFSTEM > 1 )
begin
	select @O_iErrorState = 1580 /* W2 Box for Statutory Employee (W2BFSTEM) is invalid - 0 and 1 are valid values */ /*W2BFSTEM*/
	exec @iStatus = taUpdateString
		@O_iErrorState,
		@oErrString,
		@oErrString output,
		@iAddCodeErrState output
end

/* Validate Medicare Qualified Government Employee is valid */
if ( @I_vMCRQGEMP < 0 or @I_vMCRQGEMP > 1 )
begin
	select @O_iErrorState = 4088 /* Medicare Qualified Government Employee (MCRQGEMP) is invalid - 0 and 1 are valid values */ /*EMPLOYID*/ /*MCRQGEMP*/
	exec @iStatus = taUpdateString
		@O_iErrorState,
		@oErrString,
		@oErrString output,
		@iAddCodeErrState output
end

/* Validate Withhold NY Tax is valid */
if ( @I_vNYTXDiff < 0 or @I_vNYTXDiff > 1 )
begin
	select @O_iErrorState = 4089 /* Withhold NY Tax (NYTXDiff) is invalid - 0 and 1 are valid values */ /*EMPLOYID*/ /*NYTXDiff*/
	exec @iStatus = taUpdateString
		@O_iErrorState,
		@oErrString,
		@oErrString output,
		@iAddCodeErrState output
end

/* Grant */
/*Federal Filing Status*/
if ( @I_vFDFLGSTS <> '' and @I_vFDFLGSTS <> 'SINGLE' and @I_vFDFLGSTS <> 'MAR' and @I_vFDFLGSTS <> 'NRA' and @I_vFDFLGSTS <> 'HOH' and @I_vFDFLGSTS <> 'HOHHR' and @I_vFDFLGSTS <> 'MARHR' and @I_vFDFLGSTS <> 'SGLHR' and @I_vFDFLGSTS <> 'NRAHA' )
begin
	select @O_iErrorState = 4094 /* Federal Filing Status (FDFLGSTS) is invalid */ /*EMPLOYID*/ /*FDFLGSTS*/
	exec @iStatus = taUpdateString
  		@O_iErrorState,
  		@oErrString,
  		@oErrString output,
		@iAddCodeErrState output
	return (@O_iErrorState)
end

/*EIC Filing Status*/
if ( @I_vEICFLGST <> '' and @I_vEICFLGST <> 'MAR' and @I_vEICFLGST <> 'SINGLE' and @I_vEICFLGST <> 'MAR2' )
begin
	select @O_iErrorState = 4095 /* EIC Filing Status (EICFLGST) is invalid */ /*EMPLOYID*/ /*EICFLGST*/
	exec @iStatus = taUpdateString
  		@O_iErrorState,
  		@oErrString,
  		@oErrString output,
		@iAddCodeErrState output
	return (@O_iErrorState)
end
/* End Grant */

/* Validate Update if Exists is a valid value */
if ( @I_vUpdateIfExists < 0 or @I_vUpdateIfExists > 1) 
begin
	select @O_iErrorState = 3690 /* The UpdateIfExists is invalid - 0=False & 1=True */ /*EMPLOYID*/ /*UpdateIfExists*/
	exec @iStatus = taUpdateString
  		@O_iErrorState,
  		@oErrString,
  		@oErrString output,
		@iAddCodeErrState output
end

/* Validate @I_vRequesterTrx is a valid value */
if ( @I_vRequesterTrx < 0 or @I_vRequesterTrx > 1) 
begin
	select @O_iErrorState = 3691 /* The RequesterTrx parameter is invalid - 0=False & 1=True */ /*EMPLOYID*/ /*RequesterTrx*/
	exec @iStatus = taUpdateString
  		@O_iErrorState,
  		@oErrString,
  		@oErrString output,
		@iAddCodeErrState output
end

if (@O_iErrorState <> 0)
	return (@O_iErrorState)


/**********************Get the next note index*****************************/
exec @iStatus = DYNAMICS..tasmGetNextNoteIndex
	@I_sCompanyID 		= @sCompanyID,
	@I_iSQLSessionID	= 0,
        @I_noteincrement 	= 1,
	@O_mNoteIndex 		= @nNextNoteIndex output,
	@O_iErrorState		= @iGetNextNoteIdxErrState output
select @iError = @@error
if ((@iStatus <> 0) or (@iGetNextNoteIdxErrState <> 0) or (@iError <> 0))
begin
	select @oErrString = rtrim(@oErrString) + ' ' + ltrim(rtrim(@iGetNextNoteIdxErrState))
	select @O_iErrorState = 4090 /* Unable to obtain the next note index */
	exec @iStatus = taUpdateString
 		@O_iErrorState,
 		@oErrString,
 		@oErrString output,
 		@O_oErrorState output
end

/************************************* INSERT DATA ********************************************************/
if not exists(select 1 from UPR00300 where EMPLOYID = @I_vEMPLOYID)
begin
	insert UPR00300(
		    	NOTEINDX,
			EMPLOYID,
			EXMFRFED,
			FDFLGSTS,
			FEDEXMPT,
			ADFDWHDG,
			ESTFEDWH,
			STATECD,
			LOCALTAX,
			W2BF942E,
			W2BFDCSD,
			W2BFDCMP,
			W2BFLREP,
			W2BFPPLN,
			W2BFSTEM,
			MCRQGEMP,
			EICFLGST,
			NYTXDiff
			)
		select
			@nNextNoteIndex,	/* NOTEINDX */
			@I_vEMPLOYID,		/* EMPLOYID */
			0,			/* EXMFRFED */
			@I_vFDFLGSTS,		/* FDFLGSTS */
			@I_vFEDEXMPT,		/* FEDEXMPT */
			@I_vADFDWHDG,		/* ADFDWHDG */
			@I_vESTFEDWH,		/* ESTFEDWH */
			@I_vSTATECD,		/* STATECD */
			@I_vLOCALTAX,		/* LOCALTAX */
			@I_vW2BF942E,		/* W2BF942E */
			@I_vW2BFDCSD,		/* W2BFDCSD */
			@I_vW2BFDCMP,		/* W2BFDCMP */
			@I_vW2BFLREP,		/* W2BFLREP */
			@I_vW2BFPPLN,		/* W2BFPPLN */
			@I_vW2BFSTEM,		/* W2BFSTEM */
			@I_vMCRQGEMP,		/* MCRQGEMP */
			@I_vEICFLGST,		/* EICFLGST */
			@I_vNYTXDiff		/* NYTXDiff */
	if (@@error <> 0)
	begin
		select @O_iErrorState = 4091 /* Unable to insert into the Payroll Tax Information Master Table - UPR00300 */ /*EMPLOYID*/ 
		exec @iStatus = taUpdateString
			@O_iErrorState,
			@oErrString,
			@oErrString output,
			@iAddCodeErrState output
	end
end
else
if ( @I_vUpdateIfExists = 1 )
begin
	update UPR00300 set
			FDFLGSTS = @I_vFDFLGSTS,
			FEDEXMPT = @I_vFEDEXMPT,
			ADFDWHDG = @I_vADFDWHDG,
			ESTFEDWH = @I_vESTFEDWH,
			STATECD = @I_vSTATECD,
			LOCALTAX = @I_vLOCALTAX,
			W2BF942E = @I_vW2BF942E,
			W2BFDCSD = @I_vW2BFDCSD,
			W2BFDCMP = @I_vW2BFDCMP,
			W2BFLREP = @I_vW2BFLREP,
			W2BFPPLN = @I_vW2BFPPLN,
			W2BFSTEM = @I_vW2BFSTEM,
			MCRQGEMP = @I_vMCRQGEMP,
			EICFLGST = @I_vEICFLGST,
			NYTXDiff = @I_vNYTXDiff
	where EMPLOYID = @I_vEMPLOYID
	if (@@error <> 0)
	begin
		select @O_iErrorState = 4092 /* Unable to update the Payroll Tax Information Master Table - UPR00300 */ /*EMPLOYID*/ 
		exec @iStatus = taUpdateString
			@O_iErrorState,
			@oErrString,
			@oErrString output,
			@iAddCodeErrState output
	end
end

/*********************** Call Post Proc *****************************************************/
exec @iStatus = taCreateEmployeeTaxPost
	@I_vEMPLOYID,
	@I_vFDFLGSTS,
	@I_vFEDEXMPT,
	@I_vADFDWHDG,
	@I_vESTFEDWH,
	@I_vSTATECD,
	@I_vLOCALTAX,
	@I_vW2BF942E,
	@I_vW2BFDCSD,
	@I_vW2BFDCMP,
	@I_vW2BFLREP,
	@I_vW2BFPPLN,
	@I_vW2BFSTEM,
	@I_vMCRQGEMP,
	@I_vEICFLGST,
	@I_vNYTXDiff,
	@I_vUpdateIfExists,
	@I_vRequesterTrx,
	@I_vUSRDEFND1,
	@I_vUSRDEFND2,
	@I_vUSRDEFND3,
	@I_vUSRDEFND4,
	@I_vUSRDEFND5,
	@O_iErrorState = @iCustomState output,
	@oErrString = @oErrString /*@iCustomErrString*/ output
select @iError = @@error
if @iStatus = 0 and @iError <> 0
begin
	select @iStatus = @iError
end
if (@iStatus <> 0) or (@iCustomState <> 0)
begin
	select @O_iErrorState = 4093 /* Post Custom Business Logic for taCreateEmployeeTaxPost returned an error value */ /*EMPLOYID*/ 
	exec @iStatus = taUpdateString
 		@O_iErrorState,
 		@oErrString,
 		@oErrString output,
 		@O_oErrorState output
	return (@O_iErrorState)
end
[bookmark: _GoBack]
return (@O_iErrorState)
go

grant execute on dbo.taCreateEmployeeTax to DYNGRP
go

/* End_Procs taCreateEmployeeTax */

